

# Processeur Mono-Cycle: Simulation VHDL

## Project Report

Stu. KANG Jiale , XIE Shifeng  
No. 19022100087 , 19022100008

April 29, 2022



西安电子科技大学  
XIDIAN UNIVERSITY



POLYTECH<sup>®</sup>  
SORBONNE



SORBONNE  
UNIVERSITÉ

Report submitted for

**Electronique numérique: VHDL**

at the

School of Electronic Engineering, Xidian University,  
Department of Electronics and Computer Sciences, Polytech Sorbonne.

Project Area: **Digital Electronics**

Project Supervisor: **Yann DOUZE , ZONG Ru**

### **Abstract**

The objective of this project is to design and simulate a processor. This processor will be designed from basic blocks (registers, multiplexers, memory, ALU, ...) which will be combined to produce the different blocks of the system (processing unit, instruction management unit, control unit).

The processor will be validated by simulating the execution of a simple test program. For each block, it will be described in behavioral and simulated in language VHDL with Modelsim by developing a test bench.

**Key Words:** ARM architecture, VHDL, simulation, processor

# Contents

<b>1</b>	<b>Processing Unit (Unité de Traitement)</b>	<b>3</b>
1.1	Arithmetic Logic Unit (Unité Arithmétique Logique) . . . . .	3
1.2	Register File (Banc de Registres) . . . . .	4
1.2.1	Design Register File . . . . .	4
1.2.2	Assemble ALU and Register File . . . . .	6
1.3	2 to 1 Multiplexer (Multiplexeur 2 vers 1) . . . . .	7
1.4	Sign Extension (Extension de Signe) . . . . .	8
1.5	Data Memory (Mémoire de Données) . . . . .	9
1.6	Assemble Processing Unit (Assemblage Unité de Traitement) . . . . .	10
<b>2</b>	<b>Instruction Management Unit (Unité de gestion des instructions)</b>	<b>12</b>
2.1	Design Instruction Management Unit . . . . .	12
2.2	Simulation for Instruction Management Unit . . . . .	14
<b>3</b>	<b>Control Unit (Unité de Contrôle)</b>	<b>16</b>
3.1	32-bit Register with Load Instruction (Registre 32-bit avec Commande de Charge- ment) . . . . .	16
3.2	Instruction Decoder (Decodeur d'Instructions) . . . . .	17
<b>4</b>	<b>Assembly and Validation of Processor (Assemblage et Validation du Pro- cesseur)</b>	<b>22</b>
4.1	Assemble Processor . . . . .	22
4.2	Simulate Processor . . . . .	25
<b>5</b>	<b>Test Processor Completely (Test Complet du Processeur)</b>	<b>27</b>
<b>6</b>	<b>Increasing the Instruction Set (Augmentation du Jeu d'Instruction)</b>	<b>28</b>
<b>A</b>	<b>Appendices</b>	<b>30</b>

# 1 Processing Unit (Unité de Traitement)

## 1.1 Arithmetic Logic Unit (Unité Arithmétique Logique)

The Arithmetic Logic Unit (ALU) performs the internal arithmetic manipulation of data in the processor. The instructions that are read and executed by the processor control the data flow between the registers and the ALU. The instructions also control the arithmetic operations performed by the ALU via the ALU's control inputs. A symbolic representation of an ALU is shown in Figure 1.




Figure 1: ALU Block Diagram

Where **A** and **B** are 32 bits input buses, **S** is a 32 bits output bus, **OP** is a 2 bits command signal, and **Status** is a Output Signal represent NVCZ (we just consider N here, so it is 1 bit).

For the operation of ALU, we can see from Table 1.

Thus, we set the inputs and outputs **ports** of the **entity** as follow:

```

1 entity ALU is
2     port(
3         op: in std_logic_vector(1 downto 0);
4         a,b: in std_logic_vector(31 downto 0);
5         s: out std_logic_vector(31 downto 0);
6         n: out std_logic
7     );
8 end entity;
```

Whenever instructed by the processor, the ALU performs an operation (we consider addition and subtraction, and the detailed table will be given below) on one or more values. These values, called operands, are typically obtained from two registers, or from one register and a memory location. The result of the operation is then placed back into a given destination register or memory location. The status outputs indicate any special attributes about the operation, such as whether the result was zero, negative, or if an overflow or carry occurred. [1]

Table 1: Operation Table

OP	S	Remark
00	S=A+B	ADD
01	S=B	B
10	S=A-B	SUB
11	S=A	A

Therefore, we build the architecture of ALU in **ALU.vhd**.

```

1 architecture behav of ALU is
2     signal sign: std_logic_vector(31 downto 0);
3 begin
4
5     process (a,b,op)
6     begin
7         case op is
8             when "00" => sign <=std_logic_vector(signed(a)+signed(b));
9             when "01" => sign <= b;
10            when "10" => sign <=std_logic_vector(signed(a)-signed(b));
11            when "11" => sign <= a;
12            when others => sign <= a;
13        end case ;
14    end process;
15
16    N <= sign(31);
17    s <= sign;
18
19 end architecture;
```

## 1.2 Register File (Banc de Registres)

### 1.2.1 Design Register File

Registers are temporary storage locations inside the CPU that hold data and addresses.

The register file is the component that contains all the general purpose registers of the microprocessor. A few CPUs also place special registers such as the PC and the status register in the register file. Other CPUs keep them separate.[2]

A symbolic representation of an Register File is shown in Figure 2.

Where **Clk** is a clock Signal; **Rst** is a asynchrone reset signal which active at high level; **WE** is a enable signal of writing datas and **Rw** is a 4 bits address bus of writing register; **W**, **A** and **B** are 32 bits data buses; **Ra** and **Rb** are 4 bits address buses of reading register.

Thus, the ports of the entity we defined as follow:

```

1 entity Register_File is
2     port (
3         clk, rst, WE : in std_logic ;
4         Ra, Rb, Rw : in std_logic_vector(3 downto 0);
5         A, B : out std_logic_vector(31 downto 0);
6         W : in std_logic_vector(31 downto 0)
7     );
```

```
8 end entity;
```




Figure 2: Register File Block Diagram

When WE= 1, which means *enable to write*, so that we need to write data from bus W to the register on the address Rw; And when WE= 0, we will do nothing. As for reading, it will be done in a combinatorial and simultaneous way.

Therefore, we build the architecture of Register File in **Registre\_File.vhd**.

```
1 architecture behav of Register_File is
2     type matrix is array(15 downto 0) of std_logic_vector(31 downto 0);
3     function init_banc return matrix is
4         variable result : matrix;
5     begin
6         for i in 14 downto 0 loop
7             result(i) := (others=>'0');
8         end loop;
9         result(15):=X"00000030";
10        return result;
11    end init_banc;
12
13    signal Banc: matrix:=init_banc;
14 begin
15
16    A <= Banc(to_integer(unsigned(Ra)));
17    B <= Banc(to_integer(unsigned(Rb)));
18
19    process(clk, rst)
20    begin
21        if rst = '1' then
22            for i in 14 downto 0 loop
23                Banc(i) <= (others=>'0');
24            end loop;
25            Banc(15)<=X"00000030";
26
27        elsif rising_edge(clk) then
28            if WE = '1' then
29                Banc(to_integer(unsigned(Rw)))<=W;
30            end if;
```

```

31     end if;
32     end process;
33
34 end architecture;

```

### 1.2.2 Assemble ALU and Register File

We will assemble these two component we have already finished as Figure 3.




Figure 3: Assemble ALU and Register File Block Diagram

The entity and architecture of `Assemble_ALU_and_Register_File.vhd` shows below.

```

1  entity Assemble_ALU_and_Register_File is
2      port (
3          clk, rst, WE : in std_logic ;
4          Ra, Rb, Rw : in std_logic_vector(3 downto 0);
5          W : out std_logic_vector(31 downto 0);
6          N : out std_logic;
7          op : in std_logic_vector(1 downto 0)
8      );
9  end entity;
10
11 architecture behave of Assemble_ALU_and_Register_File is
12     signal busA, busB, busW : std_logic_vector(31 downto 0);
13 begin
14
15     Register_File : entity work.Register_File port map(clk=>clk, rst=>rst, WE
16         =>We, Ra=>Ra, Rb=>Rb, Rw=>Rw, A=>busA, B=>busB, W=>busW);
17
18     ALU : entity work.ALU port map(a => busA, b=> busB, s => busW, op => op, n
19         => n);
20
21     W <= busw;
22 end architecture;

```

And based on the textbench **Assemble\_ALU\_and\_Register\_File\_tb.vhd** and command file **Assemble\_ALU\_and\_Register\_File\_test.do**, we test some operations:

$$\begin{aligned} R(1) &= R(15) \\ R(1) &= R(1) + R(15) \\ R(2) &= R(1) + R(15) \\ R(3) &= R(1) - R(15) \\ R(5) &= R(7) - R(15) \end{aligned}$$

The simulation result is shown in Figure 4. Detailed waves can be found as Figure 19 in Appendices A.




Figure 4: Simulation Waves of Assemble ALU and Regist\_File

### 1.3 2 to 1 Multiplexer (Multiplexeur 2 vers 1)

This multiplexer has a generic parameter  $N$  fixing the size of the data input and output. A symbolic representation of the multiplexer is shown in Figure 5.




Figure 5: 2 to 1 Multiplexer

Where  $A$  and  $B$  are data inputs,  $S$  is data output, and they are all  $N$ -bit;  $COM$  is a choose signal which is 1 bit. The choose table is as below.

So we build the component **MUX** in **MUX.vhd**.



Table 2: MUX Choose Table

COM	S
0	S=A
1	S=B

```

1  entity MUX is
2      generic (N : positive :=32);
3
4      port (
5          S : out std_logic_vector(N-1 downto 0);
6          A, B : in std_logic_vector(N-1 downto 0);
7          COM : in std_logic
8      );
9  end entity MUX;
10
11 architecture behave of MUX is
12 begin
13
14     process(A,B,COM)
15     begin
16         if COM = '0' then S <= A;
17         elsif COM='1' then S <= B;
18         end if;
19     end process;
20
21 end architecture;
```

## 1.4 Sign Extension (Extension de Signe)

This module is used to extend the sign of an input coded on `N bits` to `32 bits`. It therefore has a generic parameter fixing the value of `N`. A symbolic representation of the multiplexer is shown in Figure 6.



Figure 6: Sign Extension Block Diagram

Where `E` is a `N-bit` data input bus and `S` is a `32-bit` output bus.  
And part of its code in **Sign\_Extension.vhd** shows below.

```

1  entity Sign_Extension is
2      generic (N : positive :=8);
3      port (
4          S : out std_logic_vector(31 downto 0);
5          E : in std_logic_vector(N-1 downto 0)
```

```

6   );
7   end entity;
8
9   architecture behav of Sign_Extension is
10  begin
11
12      process(E)
13      begin
14          S(N-1 downto 0) <= E;
15          S(31 downto N) <= (others => E(N-1));
16      end process;
17
18  end architecture;

```

## 1.5 Data Memory (Mémoire de Données)

This memory is used to load and store 64 32-bit words. A symbolic representation of the Memory is shown in Figure 7.




Figure 7: Data Memory Block Diagram

Where **Clk** is a clock Signal; **Rst** is a asynchrone reset signal which active at high level; **WrEn** is a enable signal of writing datas and **Addr** is a 6 bits address bus; **Data In** and **Data Out** are 32 bits data buses;

Thus, the ports of the entity we defined as follow:

```

1   entity Data_Memory is
2       port (
3           clk, rst, WE : in std_logic ;
4           Addr : in std_logic_vector(5 downto 0);
5           DataOut : out std_logic_vector(31 downto 0);
6           DataIn : in std_logic_vector(31 downto 0)
7       );
8   end entity;

```

Similar to Register File, when **WrEn**= 1, it will write data from **Data In** to the **Addr** register; and when **WrEn**= 0, it will do nothing. As for reading, it will be done in a combinatorial and simultaneous way.

Therefore, we build the architecture of Data Memory in **Data\_Memory.vhd**

```

1  architecture behave of Data_Memory is
2      type matrix is array(63 downto 0) of std_logic_vector(31 downto 0);
3      signal datas: matrix;
4  begin
5
6      DataOut <= datas(to_integer(unsigned(Addr)));
7      process(clk, rst)
8      begin
9          if rst = '1' then
10             for i in 63 downto 0 loop
11                 datas(i) <= std_logic_vector(to_unsigned(i,32));
12             end loop;
13         elsif rising_edge(clk) then
14             if WE = '1' then
15                 datas(to_integer(unsigned(Addr))) <= DataIn;
16             end if;
17         end if;
18     end process;
19
20 end architecture;

```

## 1.6 Assemble Processing Unit (Assemblage Unité de Traitement)

We assemble all the components we have finished before to make a Processing Unit. All the signals are readable from the block diagram shown as Figure 8. And we give the part of source code of **Assemble\_Processing\_Unit.vhd**.




Figure 8: Assemble Processing Unit Block Diagram

```

1  architecture behave of Assemble_Processing_Unit is

```

```
2    signal busA, busB, ALUS, busExtension, busMux, busW : std_logic_vector(31
      downto 0);
3    signal DataOut : std_logic_vector(31 downto 0);
4    begin
5
6    Register_File : entity work.Register_File port map(Clk => clk, rst => rst,
      WE=> RegWr, Ra => Ra, Rb => Rb, Rw => Rw, A => busA, B => busB, W=>
      busW);
7
8    ALU : entity work.ALU port map(A => busA, B=> busMux, S => ALUS, OP =>
      ALUctr, N => N);
9
10   Sign_Extension : entity work.Sign_Extension port map (E => Imm, S =>
      busExtension);
11
12   MUX1 : entity work.MUX port map (A => busB, B=> busExtension, S=> busMux,
      COM => ALUSrc);
13
14   MUX2 : entity work.MUX port map (A => ALUS, B=> DataOut, S=> busW, COM =>
      WrSrc);
15
16   Data_Memory : entity work.Data_Memory port map (Clk => clk, rst => rst,
      Addr => ALUS(5 downto 0), WE => MemWr, DataIn => busB, DataOut =>
      DataOut);
17
18   W <= busw;
19
20 end architecture;
```

## 2 Instruction Management Unit (Unité de gestion des instructions)

### 2.1 Design Instruction Management Unit

The 32-bit instruction management unit possesses some properties:

- An instruction memory of 64 words of 32 bits similar to that of the processing unit.
- There is no write data bus and Write Enable.
- A 32-bit register (PC register) which has a clock and a asynchronous reset (active at high level).
- An extension unit of 24 to 32 signed bits similar to the component `Extension` described previously.

A symbolic representation of an Instruction Management Unit is shown in Figure 9.




Figure 9: Instruction Management Unit Block Diagram

Abstracted by the given block diagram, the function of the instruction management unit can be described as follow:

$$PC = \begin{cases} PC + 1 & nPCsel = 0 \\ PC + 1 + offset & nPCsel = 1 \end{cases}$$

Because the instructions given are 24-bit, we need to use component `Extension` to extend them to 32-bit. After that, we store these 32-bit instructions in memory by `Datamemory`.

The part of the source in `Instruction_Management_Unit.vhd` will be given below.

```

1 architecture behave of Instruction_Management_Unit is
2     signal PC, S : std_logic_vector(31 downto 0);
3 begin
4
5     Instruction_memory : entity work.instruction_memory port map (PC=> PC,
6         Instruction=> Instruction);
7     Sign_Extension : entity work.Sign_Extension generic map(N=> 24) port map (
8         E => Offset, S => S);
9
10    process(clk, rst)

```

```

9      begin
10         if rst = '1' then
11             PC <= (others => '0');
12         elsif rising_edge(clk) then
13             if nPCsel = '0' then
14                 PC <= std_logic_vector(unsigned(PC)+1);
15             else
16                 PC <= std_logic_vector(unsigned(PC)+1+ unsigned(S));
17             end if;
18         end if;
19     end process;
20 end architecture;

```

And here is the code `instruction_memory.vhd` in Appendices A.

```

1  library IEEE;
2  use IEEE.std_logic_1164.all;
3  use IEEE.numeric_std.all;
4
5  entity instruction_memory is
6      port(
7          PC: in std_logic_vector(31 downto 0);
8          Instruction: out std_logic_vector(31 downto 0)
9      );
10 end entity;
11
12 architecture RTL of instruction_memory is
13     type RAM64x32 is array(0 to 63) of std_logic_vector(31 downto
14         0);
15
16     function init_mem return RAM64x32 is
17         variable result : RAM64x32;
18     begin
19         for i in 63 downto 0 loop
20             result(i) := (others => '0');
21         end loop;
22
23         result(0) := x"E3A01020";-- 0x0 _main -- INSTRUCTION -- MOV R1,#0x20
24         result(1) := x"E3A02000";-- 0x1 -- MOV R2,#0x00
25         result(2) := x"E6110000";-- 0x2 _loop -- LDR R0,0(R1)
26         result(3) := x"E0822000";-- 0x3 -- ADD R2,R2,R0
27         result(4) := x"E2811001";-- 0x4 -- ADD R1,R1,#1
28         result(5) := x"E351002A";-- 0x5 -- CMP R1,0x2A
29         result(6) := x"BAFFFFFFB";-- 0x6 -- BLT loop
30         result(7) := x"E6012000";-- 0x7 -- STR R2,0(R1)
31         result(8) := x"EAFFFFF7";-- 0x8 -- BAL main
32
33         return result;
34     end init_mem;
35
36 signal mem: RAM64x32 := init_mem;
37
38 begin

```

```

36     Instruction <= mem(to_integer(unsigned(PC)));
37 end architecture;

```

## 2.2 Simulation for Instruction Management Unit

We build the testbench in order to test whether this unit work properly.

We mainly test the instruction such as

$$PC \leq PC + 1$$

$$PC \leq PC + 1 + \text{offset}$$

and change the value of `offset = {1, -1}`.

Given the code of the testbench `Instruction_Management_Unit_tb.vhd` as follow.

```

1  library ieee;
2  use ieee.std_logic_1164.all;
3  use ieee.numeric_std.all;
4
5  entity Instruction_Management_Unit_tb IS
6  end entity ;
7
8  architecture BENCH of Instruction_Management_Unit_tb is
9      signal Instruction : std_logic_vector(31 downto 0);
10     signal Clk          : std_logic := '0';
11     signal rst, nPCsel  : std_logic;
12     signal Offset       : std_logic_vector(23 downto 0);
13     signal Done         : boolean := False;
14     constant Period     : time := 20 ns;
15 begin
16
17     UUT : entity work.Instruction_Management_Unit port map(clk=>clk
18         , rst=> rst, nPCsel =>nPCsel, Offset => Offset, Instruction
19         => Instruction);
20
21     CLK <= '0' when Done else not CLK after Period / 2;
22     Rst <= '1', '0' after 5 ns;
23
24     process
25     begin
26
27         nPCsel<= '0';
28         Offset <= (others => '0'); -- PC <= PC + 1
29         wait for 20 ns;
30
31         nPCsel<= '0';
32         Offset <= (others => '0'); -- PC <= PC + 1
33         wait for 20 ns;
34
35         nPCsel<= '1';
36         Offset <= x"000001"; -- PC <= PC + 1 + Offset 1
37         wait for 20 ns;

```

```

36
37     nPCsel <= '0';
38     Offset <= (others => '0'); -- PC <= PC + 1
39     wait for 20 ns;
40
41     nPCsel <= '1';
42     Offset <= x"FFFFFF"; -- PC <= PC + 1 + Offset -1
43     wait for 20 ns;
44
45     nPCsel <= '0';
46     Offset <= (others => '0'); -- PC <= PC + 1
47     wait for 20 ns;
48
49     Done <= True;
50     wait;
51
52 end process;
53
54 end architecture;

```

With the command file **Instruction\_Management\_Unit\_test.do**, the waves of the simulation are shown as Figure 10. Detailed waves can be found as Figure 20 in Appendices A.




Figure 10: Simulation of Instruction Management Unit

It can be seen clearly that PC change itself as we expected.



### 3 Control Unit (Unité de Contrôle)

The control unit consists of a 32-bit register and a combinatorial decoder.

#### 3.1 32-bit Register with Load Instruction (Registre 32-bit avec Commande de Chargement)

This register will be used to store the state of the processor (Processor State Register, PSR). In this project, we only consider the state which will be limited to the value of the N flag of the ALU. A symbolic representation of this 32-bit register is shown in Figure 11.




Figure 11: 32-bit Register Block Diagram

Where **DataIn** and **DataOut** are the 32-bit buses for instruction input and output respectively, **WE** is enable signal for charge command.

Based on the analysis above, we can draw the equation:

$$\text{DataOut} = \begin{cases} \text{DataOut} & \text{WE} = 0 \\ \text{DataIn} & \text{WE} = 1 \end{cases}$$

It's not hard to synthesize the code. We show the part of the code of this component in **Registre\_Charge.vhd**.

```

1 architecture behave of Register_Charge is
2 begin
3
4     process(clk, rst)
5     begin
6         if rst = '1' then
7             DataOut <= (others => '0');
8         elsif rising_edge(clk) then
9             if WE = '1' then
10                 DataOut <= DataIn;
11             end if;
12         end if;
13     end process;
14
15 end architecture;
```

### 3.2 Instruction Decoder (Decodeur d'Instructions)

This combinatorial module generates the control signals for the processing unit, the instruction management unit, as well as the PSR register(32-bit register), which all described previously.

A symbolic representation of decoder is shown in Figure 12.




Figure 12: Decoder Block Diagram

The values of these commands depend on the statement retrieved from the instruction memory, and possibly the state of the PSR register. The structure binary of the different instructions is described in the Appendix, and we completed it as Table 3.

Table 3: Commands

INSTRUCTION	nPCSel	RegWr	ALUSrc	ALUCtr	PSREn	MemWr	WrSrc	RegSel
ADDi	0	1	1	00	0	0	0	0
ADDr	0	1	0	00	0	0	0	0
BAL	1	0	0	00	0	0	0	0
BLT	0	0	0	00	0	0	0	0
CMP	0	0	1	10	1	0	0	0
LDR	0	1	1	00	0	0	1	0
MOV	0	1	1	10	0	0	0	0
STR	0	0	1	00	0	1	0	1

The ARM instruction set formats are shown in the Appendices A. And we focus on the instructions such as

```

1  LDR Rd, [Rn, #Offset]    @ LDR (Immediate)
2  STR Rd, [Rn, #Offset]    @ STD (Immediate)
3  B    label               @ B (Always)
4  BLT  label               @ B (If Less Than)

```

These instruction set formats are shown as Figure 13, and more detailed information will be shown in the Appendices A.




Figure 13: Load, Store and Branch Instruction Set Formats

The single data transfer instructions are used to load or store single bytes or words of data. The memory address used in the transfer is calculated by adding an offset to or subtracting an offset from a base register.

The result of this calculation may be written back into the base register if auto-indexing is required.

Branch instructions contain a signed 2's complement 24 bit offset. This is shifted left two bits, sign extended to 32 bits, and added to the PC. The instruction can therefore specify a branch of +/- 32Mbytes. The branch offset must take account of the prefetch operation, which causes the PC to be 2 words (8 bytes) ahead of the current instruction. Branches beyond +/- 32Mbytes must use an offset or absolute destination which has been previously loaded into a register. In this case the PC should be manually saved in R14 if a Branch with Link type operation is required.

Thus, for these 4 instructions, bit assignments are as follow:

Table 4: LDR (Immediate) Bit Assignment

31	30	29	28	27	26	25	24	23	22	21	20	19	...	16	15	...	12	11	...	0
1	1	1	0	0	1	1	0	0	0	0	1	Rn			Rd			Offset		

Table 5: STR (Immediate) Bit Assignment

31	30	29	28	27	26	25	24	23	22	21	20	19	...	16	15	...	12	11	...	0
1	1	1	0	0	1	1	0	0	0	0	0	Rn			Rd			Offset		

Table 6: B (Always) Bit Assignment

31	30	29	28	27	26	25	24	23	...	...	...	...	...	...	...	...	...	...	...	0
1	1	1	0	1	0	1	0	Offset												

Table 7: B (If Less Than) Bit Assignment

31	30	29	28	27	26	25	24	23	...	...	...	...	...	...	...	...	...	...	...	0
1	0	1	1	1	0	1	0	Offset												

Based on the analysis above, we build the code of **Decoder.vhd**.

```

1  entity Decoder  is
2      port(
3          Instruction, PSRout : in std_logic_vector(31 downto 0);
4          Offset : out std_logic_vector(23 downto 0);
5          Immediate : out std_logic_vector(7 downto 0);
6          Rn, Rm, Rd : out std_logic_vector(3 downto 0);
7          ALUctr : out std_logic_vector(1 downto 0);
8          nPCsel, RegWr, ALUsrc, PSRen, MemWr, WrSrc, RegSel : out
              std_logic
9      );
10 end entity;
11
12 architecture behave of Decoder  is
13     type enum_instruction is (MOV, ADDi, ADDr, CMP, LDR, STR, BAL,
14                               BLT, XXX);
15     signal instr_courante: enum_instruction;
16 begin
17     Immediate <= Instruction(7 downto 0);
18     Offset    <= Instruction(23 downto 0);
19     Rn        <= Instruction(19 downto 16);
20     Rd        <= Instruction(15 downto 12);
21     Rm        <= Instruction(3  downto 0);
22     process(Instruction)
23     begin
24         case Instruction(27 downto 26) is
25             when "00"    =>
26                 case Instruction(25 downto 23) is
27                     when "001"    => instr_courante <= ADDr;
28                     when "101"    =>
29                         case Instruction(29) is
30                             when '1'    => instr_courante <= ADDi;
31                             when others => instr_courante <= XXX;
32                         end case;
33                     when "110" | "010" => instr_courante <= CMP;
34                     when "111"    => instr_courante <= MOV;
35                     when others    => instr_courante <= XXX;
36                 end case;
37
38             when "01"    =>
39                 case Instruction(20) is
40                     when '0'    =>
41                         case Instruction(29) is
42                             when '1'    => instr_courante <= STR;
43                             when others => instr_courante <= XXX;
44                         end case;
45                     when '1'    => instr_courante <= LDR;
46                     when others => instr_courante <= XXX;
47                 end case;

```

```

48
49         when "10"      =>
50             case Instruction(29 downto 28) is
51                 when "10"      => instr_courante <= BAL;
52                 when "11"      => instr_courante <= BLT;
53                 when others    => instr_courante <= XXX;
54             end case;
55         when others => instr_courante <= XXX;
56     end case;
57 end process;
58
59 process(instr_courante)
60 begin
61     -- MemWr et RegSel
62     case instr_courante is
63         when STR      => MemWr   <= '1';
64                        ALUSrc  <= '1';
65                        RegSel  <= '1';
66         when others  => MemWr   <= '0';
67                        RegSel  <= '0';
68     end case;
69
70     -- WrSrc
71     case instr_courante is
72         when LDR      => WrSrc <= '1';
73         when others  => WrSrc <= '0';
74     end case;
75
76     -- PSRen et UALctr
77     case instr_courante is
78         when CMP      => PSRen  <= '1';
79                        ALUctr <= "10";
80         when MOV      => PSRen  <= '0';
81                        ALUctr <= "01";
82         when others  => PSRen  <= '0';
83                        ALUctr <= "00";
84     end case;
85
86     --UALsrc
87     case instr_courante is
88         when ADDr     => ALUsrc <= '0';
89         when CMP      => ALUsrc <= Instruction(25);
90         when others  => ALUsrc <= '1';
91     end case;
92
93     -- RegWr
94     case instr_courante is
95         when ADDi | ADDr | LDR | MOV => RegWr <= '1';
96         when others                    => RegWr <= '0';
97     end case;

```

```
98
99      -- nPCsel
100      case instr_courante is
101          when BAL      => nPCsel <= '1';
102          when BLT      => nPCsel <= PSRout(31);
103          when others   => nPCsel <= '0';
104      end case;
105
106      end process;
107
108  end architecture;
```

At this point, all the components have been constructed. In the next part, we will try to assemble the processor by using these components.

## 4 Assembly and Validation of Processor (Assemblage et Validation du Processeur)

### 4.1 Assemble Processor

We modifying the modeling of the processor by assembling its three main units

- The instruction management unit
- The processing unit
- The control unit

Complete the previously designed processing unit by adding a 2-input 4-bit multiplexer controlled by the RegSel control signal generated by the control unit. This multiplexer will be placed at the input of the address Rb of the register file, as shown in the Figure 14.

According to this block diagram, we modify the code for **Processing\_Unit.vhd**.

```

1  entity Processing_Unit is
2      port (
3          clk, rst, MemWr, RegWr, ALUSrc, WrSrc ,RegSel, PSREn: in std_logic ;
4          Rn, Rm, Rd : in std_logic_vector(3 downto 0);
5          busout : out std_logic_vector(31 downto 0);--PSR[31..0]
6          Imm : in std_logic_vector(7 downto 0);
7          ALUctr : in std_logic_vector(1 downto 0)
8      );
9  end entity;
10
11 architecture behave of Processing_Unit is
12     signal busA,busB,ALUS,busExtension,busMux, busW : std_logic_vector(31
13         downto 0);
14     signal DataOut ,fl: std_logic_vector(31 downto 0);
15     signal Rb : std_logic_vector(3 downto 0);
16     signal N: std_logic;
17 begin
18     Register_File : entity work.Register_File port map(Clk => clk, rst => rst,
19         WE=> RegWr, Ra => Rn, Rb => Rb, Rw => Rd, A => busA, B => busB, W=>
20         busW);
21
22     ALU : entity work.ALU port map(A => busA, B=> busMux, S => ALUS, OP =>
23         ALUctr, N => N);
24
25     Sign_Extension : entity work.Sign_Extension port map ( E => Imm, S =>
26         busExtension);
27
28     MUX1 : entity work.MUX port map (A => busB, B=> busExtension, S=> busMux,
29         COM => ALUSrc);
30
31     MUX2 : entity work.MUX port map (A => ALUS, B=> DataOut, S=> busW, COM =>
32         WrSrc);

```




Figure 14: Processing Unit Block Diagram



```

28   Data_Memory : entity work.Data_Memory port map (Clk => clk, rst => rst,
29           Addr => ALUS(5 downto 0), WE => MemWr, DataIn => busB, DataOut =>
30           DataOut);
31
32   MUX3 : entity work.MUX generic map( N=> 4) port map (A => Rm, B => Rd, COM
33           => RegSel, S=> Rb);
34
35   Register_Charge : entity work.Register_Charge port map (clk => clk, rst =>
36           rst, WE => PSREn, DataIn => f1, DataOut => busout);
37
38   f1 <= N&"000"&X"00000000";
39
40 end architecture;

```

With addition of **Processing\_Unit**, **Instruction\_Management\_Unit** and **Decoder**, the **Processor** can be assembled as Figure 15. The detailed block diagram will be given in the Appendices A.




Figure 15: ProcessorBlock Diagram

Thus, the core code for **Processor.vhd** is shown below.

```

1 architecture behave of Processor is
2     signal nPCSel, MemWr, RegWr, ALUSrc, WrSrc, RegSel, PSRen : std_logic ;
3     signal ALUctr : std_logic_vector(1 downto 0);
4     signal offset : std_logic_vector(23 downto 0);
5     signal Immediate: std_logic_vector(7 downto 0);
6     signal Instruction, busout : std_logic_vector(31 downto 0);--busout -> PSR
7     signal Rn, Rd, Rm : std_logic_vector(3 downto 0);
8 begin
9
10    Processing_Unit : entity work.Processing_Unit port map(clk => clk, rst =>
11        rst, RegWr=> RegWr, Rn => Rn, Rd => Rd, Rm=> Rm, busout => busout, Imm
12        => Immediate, ALUctr=> ALUctr, MemWr=> MemWr, ALUSrc=> ALUSrc,WrSrc=>
13        WrSrc, RegSel=> RegSel, PSRen => PSRen);
14
15    Instruction_Management_Unit : entity work.Instruction_Management_Unit port
16        map(Clk => clk, rst => rst, nPCSel=> nPCSel, Instruction=> Instruction
17        , offset => offset);
18
19    Decoder : entity work.Decoder port map(RegWr=> RegWr, Rn => Rn, Rd => Rd,
20        Rm=> Rm, psrout => busout, Immediate =>Immediate, ALUctr=> ALUctr,
21        MemWr=> MemWr, ALUSrc=> ALUSrc, WrSrc=> WrSrc, RegSel=> RegSel, PSRen
22        => PSRen, nPCSel=> nPCSel, Instruction=> Instruction, offset=> offset);
23
24 end architecture;

```

## 4.2 Simulate Processor

According to the testbench shown below, we run the simulation with command file **Processor\_test.do** and obtain the waves as Figure 16. Detailed waves can be found as Figure 25 in Appendices A.

```

1 library ieee;
2 use ieee.std_logic_1164.all;
3 use ieee.numeric_std.all;
4
5 entity Processor_tb is
6 end entity;
7
8 architecture test_bench of Processor_tb is
9     signal clk, rst: std_logic;
10    signal done : std_logic := '0';
11    constant clk_period : time:= 10 ns;
12 begin
13
14    UUT : entity work.Processor(behave)port map(clk => clk,rst => rst);
15
16    rst <= '1', '0' after clk_period;
17
18    clock : process is
19 begin

```

```
38 end architecture;
```




Figure 16: Simulation of Processing Unit

## 5 Test Processor Completely (Test Complet du Processeur)

In this part, we will test the processor complement.

We first change the code Assembly given to the instruction code in **instruction\_memory2.vhd** according to Figure 21.

```

1 result (0):=x"E3A00010"; -- 0x0 _main -- MOV R0,#0x10
2 result (1):=x"E3A01001"; -- 0x1 -- MOV R1,#0x01
3 result (2):=x"E6103000"; -- 0x2 _for -- LDR R3,[R0]
4 result (3):=x"E6104001"; -- 0x3 -- LDR R4,[R0,#1]
5 result (4):=x"E6004000"; -- 0x4 -- STR R4,[R0]
6 result (5):=x"E6003001"; -- 0x5 -- STR R3,[R0,#1]
7 result (6):=x"E2811001"; -- 0x6 -- ADD R1,R1,#1
8 result (7):=x"E2800001"; -- 0x7 -- ADD R0,R0,#1
9 result (8):=x"E351000A"; -- 0x8 -- CMP R1,0xA
10 result (9):=x"BAFFFFFF8"; -- 0x9 -- BLT loop
11 result (10):=x"EAffFFFF"; -- 0xA _wait -- BAL wait

```

After running the simulation with command file **Processor\_test.do**, we obtaine waves as Figure 17. Detailed waves can be found as Figure 26 in Appendices A.




Figure 17: Simulation Waves of Processing Unit Completely

## 6 Increasing the Instruction Set (Augmentation du Jeu d'Instruction)

In this section, we have added additional command suffix support, including EQ, NE, LT, GT. The detailed information is given in Table 8.

Table 8: Condtion Code			
Code	Suffix	Flag	Meaning
0000	EQ	Z=1	equal
0001	NE	Z=0	not equal
1011	LT	N=1	less than
1100	GT	N=0	greater than

In order to match these changes, we need to modify the codes we have done before such as **ALU.vhd**, **Decoder.vhd** and **Data\_memory.vhd** and rewrite the instructions in **instruction\_memory3.vhd**.

For **ALU.vhd**, we add an additional output port Z which indicated ZERO in Status.

For **Decoder.vhd**, we add the branch of **case** to make it decoder EQ, NE, LT, GT successfully according to Table 8.

For **Data\_memory.vhd**, we add the initialization for datas in memory as Table 9.

Table 9: Datas in Memory

Address	Data
0x20	3
0x21	107
0x22	27
0x23	12
0x24	322
0x25	155
0x27	63

Finally, we created the command in **instruction\_memory3.vhd** as below.

```

1 result (0) :=x"E3A00020";-- 0x0 _start -- MOV R0,#0x20
2 result (1) :=x"E3A02001";-- 0x1 -- MOV R2,#1
3 result (2) :=x"E3A02000";-- 0x2 _while -- MOV R2,#0
4 result (3) :=x"E3A01001";-- 0x3 -- MOV R1,#1
5 result (4) :=x"E6103000";-- 0x4 _for -- LDR R3,[R0]
6 result (5) :=x"E6104001";-- 0x5 -- LDR R4,R0,#1
7 result (6) :=x"E1530004";-- 0x6 -- CMP R3, R4
8 result (7) :=x"C6004000";-- 0x7 -- STRGT R4,[R0]
9 result (8) :=x"C6003001";-- 0x8 -- STRGT R3,[R0,#1]
10 result (9) :=x"C2822001";-- 0x9 -- ADDGT R2,R2,#1
11 result (10):=x"E2800001";-- 0xA -- ADD R0, R0, #1
12 result (11):=x"E2811001";-- 0xB -- ADD R1, R1, #1
13 result (12):=x"E3510007";-- 0xC -- CMP R1, #0x07
14 result (13):=x"BAFFFFFF6";-- 0xD -- BLT FOR
15 result (14):=x"E3520000";-- 0xE -- CMP R2, #0

```

```

16 result (15):=x"E3A00020";-- 0xF          -- MOV R0, #0x20
17 result (16):=x"1AFFFFF1";-- 0x10        -- BNE WHILE
18 result (17):=x"EAFFFFFF";-- 0x11 _wait  -- BAL wait

```

After running the testbench with command file **Processor\_test.do**, we obtain the waves as Figure 18.




Figure 18: Simulation Waves of Processing Unit with IS Increasing

## A Appendices

In Section 1.2.2, the full view of the waves in simulation **Assemble\_\_ALU\_\_and\_\_Regist\_\_File\_\_tb.v** is shown as Figure 19.

In Section 2.2, the full view of the waves in simulation **Instruction\_\_Management\_\_Unit\_\_tb.vhd** is shown as Figure 20.

In Section 3.2, the ARM instruction set formats are shown as Figure 21.

And in the same section (Section 3.2), the detailed information about **Single Data Transfer** and **Branch** are shown as Figure 22 and Figure 23.

In Section 4.1, the Block Diagram of Processor as Figure 24.

In Section 4.2, the full view of the waves in simulation **Processor\_\_tb.vhd** in **part 4** is shown as Figure 25.

In Section 5, the full view of the waves in simulation in **part 5** is shown as Figure 26.

In Section 6, the full view of the waves in simulation in **part 6** is shown as Figure 27.




Figure 19: Simulation Waves of Assemble ALU and Regist\_File






Figure 20: Simulation of Instruction Management Unit

3 3 2 2 2 2 2 2 2 2 2 2 1 1 1 1 1 1 1 1 1 1 9 8 7 6 5 4 3 2 1 0  
1 0 9 8 7 6 5 4 3 2 1 0 9 8 7 6 5 4 3 2 1 0

Cond	0	0	I	Opcode				S	Rn				Rd				Operand 2								<i>Data Processing / PSR Transfer</i>				
Cond	0	0	0	0	0	0	A	S	Rd				Rn				Rs		1	0	0	1	Rm				<i>Multiply</i>		
Cond	0	0	0	0	1	U	A	S	RdHi				RdLo				Rn		1	0	0	1	Rm				<i>Multiply Long</i>		
Cond	0	0	0	1	0	B	0	0	Rn				Rd				0	0	0	0	1	0	0	1	Rm				<i>Single Data Swap</i>
Cond	0	0	0	1	0	0	1	0	1	1	1	1	1	1	1	1	1	1	1	0	0	0	1	Rn				<i>Branch and Exchange</i>	
Cond	0	0	0	P	U	0	W	L	Rn				Rd				0	0	0	0	1	S	H	1	Rm				<i>Halfword Data Transfer: register offset</i>
Cond	0	0	0	P	U	1	W	L	Rn				Rd				Offset				1	S	H	1	Offset				<i>Halfword Data Transfer: immediate offset</i>
Cond	0	1	I	P	U	B	W	L	Rn				Rd				Offset								<i>Single Data Transfer</i>				
Cond	0	1	1																			1					<i>Undefined</i>		
Cond	1	0	0	P	U	S	W	L	Rn				Register List												<i>Block Data Transfer</i>				
Cond	1	0	1	L	Offset																				<i>Branch</i>				
Cond	1	1	0	P	U	N	W	L	Rn				CRd				CP#				Offset								<i>Coprocessor Data Transfer</i>
Cond	1	1	1	0	CP Opc				CRn				CRd				CP#				CP		0	CRm				<i>Coprocessor Data Operation</i>	
Cond	1	1	1	0	CP Opc				L	CRn				Rd				CP#				CP		1	CRm				<i>Coprocessor Register Transfer</i>
Cond	1	1	1	1	Ignored by processor																				<i>Software Interrupt</i>				

3 3 2 2 2 2 2 2 2 2 2 2 1 1 1 1 1 1 1 1 1 1 9 8 7 6 5 4 3 2 1 0  
1 0 9 8 7 6 5 4 3 2 1 0 9 8 7 6 5 4 3 2 1 0

Figure 21: ARM Instruction Set Formats




Figure 22: Single Data Transfer Instructions




Figure 23: Branch Instructions






Figure 24: Processor Block Diagram








Entity: processor\_tb Architecture: test\_bench Date: Fri Apr 29 11:27:07 CST 2022 Row: 1 Page: 1









## References

- [1] John Catsoulis. *Designing Embedded Hardware: Create New Computers and Devices*. O'Reilly Media, Inc., 2005.
- [2] J.D. Dumas. *Computer Architecture: Fundamentals and Principles of Computer Design*. Taylor & Francis, 2005.